

## Description

Predefined styles provide an easy way to customize the look of a table. You can access predefined styles by typing

```
. collect style use stylename
```

when you create a table with `collect` or by specifying the `style(stylename)` option when you create a table with the `dtable`, `etable`, `table`, `tabulate oneway`, `tabulate twoway`, `tabulate`, `summarize()`, `svy: tabulate oneway`, or `svy: tabulate twoway` command.

You can create your own style files, which you can use over and over as you build tables. Or you can use any of the numerous style files that are installed with Stata.

## Remarks and examples

Remarks are presented under the following headings:

### Creating a new style

#### Default styles

default	etable	table
dtable	lcstats	tabulate

#### Other styles

anova	table-reg1-fv1	table-reg3-fv1
coef-table	table-reg2	table-right
table-1	table-reg2-fv1	table-tab2
table-reg1	table-reg3	

#### Targeted styles

anova_borders	default_tex	etable_stars
anova_halign	dtable_borders	lcstats_font
anova_headers	dtable_composites	lcstats_halign
anova_layout	dtable_font	lcstats_headers
anova_nformats	dtable_halign	lcstats_nformats
coef-table_halign	dtable_headers	lcstats_smcl
coef-table_headers	dtable_nformats	lcstats_txt
default_borders	etable_borders	table_cidelimiter
default_cidelimiter	etable_etable	table_headers
default_halign	etable_font	table_nformats
default_headers	etable_halign	tabulate_borders
default_margins	etable_headers	tabulate_headers
default_nformats	etable_nformats	tabulate_nformats
default_smcl	etable_showitem	

#### Modifying the default style

## Creating a new style

If you find yourself specifying the same `collect` style commands to many of your tables, you can create a style once and then apply it to many tables you create in the future.

For example, suppose you use the `table` command to produce tables of regression results, such as the ones you would get by typing

```
. use https://www.stata-press.com/data/r19/nhanes2
. table, command(regress bpsystol age weight) ///
    command(regress bpsystol age weight i.region)
```

By default, `table` will use the style named `table`, which means `table` uses the styles defined in the file `style-table.stjson`, which is installed with Stata. However, suppose that you find that you almost always want to show the values rather than the default labels for the `command` dimension, hide the titles but show the labels for other dimensions, and show the base categories for only the main effects but not the interaction terms for factor variables in the models. After each `table` command similar to the one above, you could type

```
. collect style header, title(hide) level(label)
. collect style header command, level(value)
. collect style showbase factor
. collect preview
```

to make these modifications. Alternatively, you could create your own style by typing

```
. collect style clear
. collect style use style-table, replace
. collect style header, title(hide) level(label)
. collect style header command, level(value)
. collect style showbase factor
. collect style save mytablereg, replace
```

In the first line above, we clear the current collection's style properties. In the second line, we specify that we want to start with the styles that `table` uses by default. The third through fifth lines apply the style edits that we prefer. Finally, in the last line we save our style, giving it the name `mytablereg`.

Once we have created this style, we can now use it with subsequent `table` commands. For example, we can type

```
. table, command(regress bpsystol age weight) ///
    command(regress bpsystol age weight i.region) style(mytablereg)
```

More generally, we can create a new style by typing

```
. collect clear
. collect style use basestyle, replace
. style modifications
. collect style save mystyle, replace
```

The `collect style use` command is not necessary, but it is often most convenient to start with a style that you are familiar with such as `style-table`, which is the default for the `table` command, or `style-default`, which is the default for the `collect` command, and then make edits to that style.

After you have created your new style, you can apply it to a table created by `collect` by typing

```
. collect style use mystyle
```

or by adding the `style(mystyle)` option to your `dtable`, `etable`, `table`, or `tabulate` command.

## Default styles

The following styles define the default style properties for `collect`, `dtable`, `etable`, `table`, and `tabulate`.

These styles are composed from other targeted styles and are not directly created by any `collect style` commands. Targeted styles are defined by `collect style` commands that focus on specific style properties, such as borders, headers, or formats.

### default

This style is the default for tables created by `collect`. It is defined in `style-default.stjson` and is composed as follows:

```
{
  "StyleIncludes": [
    "default_borders",
    "default_cidelimiter",
    "default_halign",
    "default_headers",
    "default_margins",
    "default_nformats",
    "default_smcl",
    "default_tex"
  ]
}
```

This style can be accessed by typing

```
. collect style use default
```

or adding the `style(default)` option to the `table` command.

You can change the default style for `collect` with `set collect_style`; see [TABLES] [set collect\\_style](#).

For an example of the default style, see the [first example](#) in [TABLES] [Example 5](#).

### dtable

This style is the default for tables created by `dtable`. It is defined in `style-dtable.stjson` and is composed as follows:

```
{
  "StyleIncludes": [
    "default_margins",
    "default_smcl",
    "default_tex",
    "dtable_borders",
    "dtable_font",
    "dtable_halign",
    "dtable_headers",
    "dtable_nformats"
  ]
}
```

This style can be accessed by typing

```
. collect style use dtable
```

or adding the `style(dtable)` option to the `dtable` command.

You can change the default style for `dtable` with `set dtable_style`; see [TABLES] [set dtable\\_style](#).

## etable

This style is the default for tables created by `etable`. It is defined in `style-etable.stjson` and is composed as follows:

```
{
  "StyleIncludes": [
    "default_cidelimiter",
    "default_margins",
    "default_smcl",
    "default_tex",
    "etable_borders",
    "etable_etable",
    "etable_font",
    "etable_halign",
    "etable_headers",
    "etable_nformats",
    "etable_showitem",
    "etable_stars"
  ]
}
```

This style can be accessed by typing

```
. collect style use etable
```

or adding the `style(etable)` option to the `etable` command.

You can change the default style for `etable` with `set etable_style`; see [TABLES] [set etable\\_style](#).

For an example of the `etable` style, see [TABLES] [Example 6](#).

## lcstats

This style is the default for tables created by `lcstats`. It is defined in `style-lcstats.stjson` and is composed as follows:

```
{
  "StyleIncludes": [
    "default_margins",
    "default_tex",
    "default_borders",
    "lcstats_font",
    "lcstats_halign",
    "lcstats_headers",
    "lcstats_nformats",
    "lcstats_smcl",
    "lcstats_txt"
  ]
}
```

This style can be accessed by typing

```
. collect style use lcstats
```

or adding the `style(lcstats)` option to the `lcstats` command.

**table**

This style is the default for tables created by `table`. It is defined in `style-table.stjson` and is composed as follows:

```
{
  "StyleIncludes": [
    "default_borders",
    "default_halign",
    "default_margins",
    "default_smcl",
    "default_tex",
    "table_cidelimiter",
    "table_headers",
    "table_nformats"
  ]
}
```

This style can be accessed by typing

```
. collect style use table
```

or adding the `style(table)` option to the `table` command.

You can change the default style for `table` with `set table_style`; see [TABLES] [set table\\_style](#).

For an example of the `table` style, see the [first example](#) in [TABLES] [Example 2](#).

**tabulate**

This style is the default for tables created by `tabulate`, `tab1`, `tab2`, `tabi`, and `svy: tabulate`. It is defined in `style-tabulate.stjson` and is composed as follows:

```
{
  "StyleIncludes": [
    "default_halign",
    "default_margins",
    "default_smcl",
    "default_tex",
    "tabulate_borders",
    "tabulate_headers",
    "tabulate_nformats"
  ]
}
```

This style can be accessed by typing

```
. collect style use tabulate
```

or adding the `style(tabulate)` option to the `tabulate` command.

You can change the default style for `tabulate` with `set tabulate_style`; see [TABLES] [set tabulate\\_style](#).

For examples of this style, see *Publish your tables* in [R] [tabulate oneway](#), [R] [tabulate twoway](#), [R] [tabulate, summarize\(\)](#), [SVY] [svy: tabulate oneway](#), and [SVY] [svy: tabulate twoway](#).

## Other styles

The following styles are installed with Stata. They are example styles intended to help you customize your tables.

### anova

This style can be used with results from `anova` or `oneway` and has the building blocks for constructing ANOVA tables. It is defined in `style-anova.stjjson` and is composed as follows:

```
{
  "StyleIncludes": [
    "anova_borders",
    "anova_halign",
    "anova_headers",
    "anova_layout",
    "anova_nformats",
    "default_margins",
    "default_smcl",
    "default_tex",
  ]
}
```

This style can be accessed by typing

```
. collect style use anova, replace
```

or, if you want to keep the collection's current style properties that do not conflict with this style, by typing

```
. collect style use anova, layout override
```

For an example of the `anova` style, see the [first example](#) in [TABLES] **Example 8**.

### coef-table

This style is useful for building tables with model coefficients. It is defined in `style-coef-table.stjjson` and is composed as follows:

```
{
  "StyleIncludes": [
    "coef-table_halign",
    "coef-table_headers",
    "default_borders",
    "default_cidelimiter",
    "default_margins",
    "default_nformats",
    "default_smcl",
    "default_tex"
  ]
}
```

This style can be accessed by typing

```
. collect style use coef-table
```

or adding the `style(coef-table)` option to the `table` command.

**table-1**

This style builds on the `table` style and has the following modifications:

1. The names of statistics, the levels of the `result` dimension, are hidden.
2. The row headers are stacked into a single column, and vertical space is added between dimensions.
3. The row headers are right-aligned.

This style is defined using the following commands:

```
. collect style use style-table, replace
. collect style header result, level(hide)
. collect style row stack, nodelimiter spacer
. collect style cell cell_type[row-header], halign(right)
```

This style is saved in `style-table-1.stjson` and can be accessed by typing

```
. collect style use table-1
```

or adding the `style(table-1)` option to the `table` command.

For an example of the `table-1` style, see [Classic Table 1](#) in [R] [table summary](#).

**table-reg1**

This style builds on the `table` style and has the following modification:

1. The level labels for the `command` dimension, the full commands typed in the `command()` option, are hidden.

This style is defined using the following commands:

```
. collect style use style-table, replace
. collect style header command, level(value)
```

This style is saved in `style-table-reg1.stjson` and can be accessed by typing

```
. collect style use table-reg1
```

or adding the `style(table-reg1)` option to the `table` command.

For an example of the `table-reg1` style, see [Regression results with factor variables](#) in [R] [table regression](#).

## table-reg1-fv1

This style builds on the `table` style and has the following modifications:

1. The dimension titles are hidden for all dimensions, and the level labels are shown for all dimensions other than `command`.
2. The level labels for the `command` dimension, the full commands typed in the `command()` option, are hidden.
3. The base category is shown for the main effects of factor variables but not for interactions.

This style is defined using the following commands:

```
. collect style use style-table, replace
. collect style header, title(hide) level(label)
. collect style header command, level(value)
. collect style showbase factor
```

This style is saved in `style-table-reg1-fv1.stjson` and can be accessed by typing

```
. collect style use table-reg1-fv1
```

or adding the `style(table-reg1-fv1)` option to the `table` command.

For an example of the `table-reg1-fv1` style, see [Regression results with factor variables](#) in [R] [table regression](#).

## table-reg2

This style builds on the `table` style and has the following modifications:

1. The level labels for the `command` dimension, the full commands typed in the `command()` option, are hidden.
2. The names of statistics, the levels of the `result` dimension, are hidden.
3. The row headers are right-aligned.

This style is defined using the following commands:

```
. collect style use style-table, replace
. collect style header command, level(value)
. collect style header result, level(hide)
. collect style cell cell_type[row-header], halign(right)
```

This style is saved in `style-table-reg2.stjson` and can be accessed by typing

```
. collect style use table-reg2
```

or adding the `style(table-reg2)` option to the `table` command.

**table-reg2-fv1**

This style builds on the `table` style and has the following modifications:

1. The dimension titles are hidden for all dimensions, and the level labels are shown for all dimensions other than `command`.
2. The level labels for the `command` dimension, the full commands typed in the `command()` option, are hidden.
3. The names of statistics, the levels of the `result` dimension, are hidden.
4. The row headers are right-aligned.
5. The base category is shown for the main effects of factor variables but not for interactions.

This style is defined using the following commands:

```
. collect style use style-table, replace
. collect style header, title(hide) level(label)
. collect style header command, level(value)
. collect style header result, level(hide)
. collect style cell cell_type[row-header], halign(right)
. collect style showbase factor
```

This style is saved in `style-table-reg2-fv1.stjson` and can be accessed by typing

```
. collect style use table-reg2-fv1
```

or adding the `style(table-reg2-fv1)` option to the `table` command.

For an example of the `table-reg2-fv1` style, see *Regression results with factor variables* in [R] [table regression](#).

### table-reg3

This style builds on the `table` style and has the following modifications:

1. The level labels for the `command` dimension, the full commands typed in the `command()` option, are hidden.
2. The names of statistics, the levels of the `result` dimension, are hidden.
3. The values in the body of the table are horizontally centered within the cells.
4. The row headers are right-aligned.
5. The row headers are stacked into a single column, and vertical space is added between dimensions.

This style is defined using the following commands:

```
. collect style use style-table, replace
. collect style header command, level(value)
. collect style header result, level(hide)
. collect style cell cell_type[item], halign(center)
. collect style cell cell_type[row-header], halign(right)
. collect style row stack, spacer
```

This style is saved in `style-table-reg3.stjson` and can be accessed by typing

```
. collect style use table-reg3
```

or adding the `style(table-reg3)` option to the `table` command.

For an example of the `table-reg3` style, see [Tables with results from a single command](#) in [R] [table regression](#).

## table-reg3-fv1

This style builds on the `table` style and has the following modifications:

1. The dimension titles are hidden for all dimensions, and the level labels are shown for all dimensions other than `command` and `result`.
2. The level labels for the `command` dimension, the full commands typed in the `command()` option, are hidden.
3. The names of statistics, the levels of the `result` dimension, are hidden.
4. The values in the body of the table are horizontally centered within the cells.
5. The row headers are right-aligned.
6. The row headers are stacked into a single column, and vertical space is added between dimensions.
7. The base category is shown for the main effects of factor variables but not for interactions.

This style is defined using the following commands:

```
. collect style use style-table, replace
. collect style header, title(hide) level(label)
. collect style header command, level(value)
. collect style header result, level(hide)
. collect style cell cell_type[item], halign(center)
. collect style cell cell_type[row-header], halign(right)
. collect style row stack, spacer
. collect style showbase factor
```

This style is saved in `style-table-reg3-fv1.stjson` and can be accessed by typing

```
. collect style use table-reg3-fv1
```

or adding the `style(table-reg3-fv1)` option to the `table` command.

## table-right

This style builds on the `table` style and has the following modification:

1. The row headers are right-aligned.

This style is defined using the following commands:

```
. collect style use style-table, replace
. collect style cell cell_type[row-header], halign(right)
```

This style is saved in `style-table-right.stjson` and can be accessed by typing

```
. collect style use table-right
```

or adding the `style(table-right)` option to the `table` command.

For an example of the `table-right` style, see [Customizing results](#) in [R] [table oneway](#).

## table-tab2

This style builds on the `table` style and has the following modifications:

1. The row headers are right-aligned.
2. The names of statistics, the levels of the `result` dimension, are hidden.
3. The row headers are stacked into a single column, and vertical space is added between dimensions.

This style is defined using the following commands:

```
. collect style use style-table, replace
. collect style cell cell_type[row-header], halign(right)
. collect style header result, level(hide)
. collect style row stack, spacer
```

This style is saved in `style-table-tab2.stjson` and can be accessed by typing

```
. collect style use table-tab2
```

or adding the `style(table-tab2)` option to the `table` command.

For an example of the `table-tab2` style, see [Customizing results](#) in [\[R\] table twoway](#).

## Targeted styles

The following styles each focus on specific style properties and are referenced by other styles.

### anova\_borders

This style defines cell border properties targeted to look like most ANOVA tables in Stata output. It is part of the definition for the `anova` style and is defined using the following commands:

```
. collect style clear
. collect style cell border_block[corner], border(bottom right)
. collect style cell border_block[row-header], border(right)
. collect style cell border_block[column-header], border(bottom)
. collect style cell term[Total], border(top)
```

This style is saved in `style-anova_borders.stjson` and can be accessed by typing

```
. collect style use anova_borders
```

as one step in building your own style.

## anova\_halign

This style defines horizontal alignment properties targeted to look like most ANOVA tables in Stata output. It is part of the definition for the `anova` style and is defined using the following commands:

```
. collect style clear
. collect style cell cell_type[row-header], halign(right)
```

This style is saved in `style-anova_halign.stjson` and can be accessed by typing

```
. collect style use anova_halign
```

as one step in building your own style.

## anova\_headers

This style defines header properties targeted to look like most ANOVA tables in Stata output. It is part of the definition for the `anova` style and is defined using the following commands:

```
. collect style clear
. collect style header, basestyle title(hide) level(label)
. collect style header term, level(value) fvlevels(hide)
. collect style column, extraspace(1) dups(center)
. collect style row stack, length(24) nobinder
```

This style is saved in `style-anova_headers.stjson` and can be accessed by typing

```
. collect style use anova_headers
```

as one step in building your own style.

## anova\_layout

This style defines the standard ANOVA layout. It is part of the definition for the `anova` style and is defined using the following commands:

```
. collect style clear
. collect layout (term) (result[SS df MS F p])
```

This style is saved in `style-anova_layout.stjson` and can be accessed by typing

```
. collect style use anova_layout
```

as one step in building your own style.

## anova\_nformats

This style defines numeric format properties. It is part of the definition for the `anova` style and is defined using the following commands:

```
. collect style clear
. collect style cell, basestyle nformat(%18.2f)
. collect style cell result[df], nformat(%18.0fc)
. collect style cell result[p], nformat(%6.4f) minimum(0.0001)
```

This style is saved in `style-anova_nformats.stjson` and can be accessed by typing

```
. collect style use anova_nformats
```

as one step in building your own style.

## coef-table\_halign

This style defines horizontal alignment properties targeted to look like Stata's coefficient and estimation tables. It is part of the definition for the `coef-table` style and is defined using the following commands:

```
. collect style clear
. collect style cell cell_type[corner], halign(center)
. collect style cell cell_type[column-header], halign(right)
. collect style cell cell_type[column-header]#result[_r_ci _r_cri], halign(center)
. collect style cell cell_type[item], halign(right)
. collect style cell cell_type[row-header]#coleq, halign(left)
. collect style cell cell_type[row-header]#colname, halign(right)
```

This style is saved in `style-coef-table_halign.stjson` and can be accessed by typing

```
. collect style use coef-table_halign
```

as one step in building your own style.

## coef-table\_headers

This style defines header properties targeted to look like Stata's coefficient and estimation tables. It is part of the definition for the `coef-table` style and is defined using the following commands:

```
. collect style clear
. collect style header, basestyle title(hide) level(label)
. collect style header stars, title(hide) level(hide)
. collect style row stack, noindent nobinder
```

This style is saved in `style-coef-table_headers.stjson` and can be accessed by typing

```
. collect style use coef-table_headers
```

as one step in building your own style.

## default\_borders

This style defines cell border properties targeted to look like most tables in Stata output. It is part of the definition for the `default`, `lcstats`, `table`, and `coef-table` styles and is defined using the following commands:

```
. collect style clear
. collect style cell border_block[corner], border(top right)
. collect style cell border_block[column-header], border(top)
. collect style cell border_block[row-header], border(top bottom right)
. collect style cell border_block[item], border(top bottom)
```

This style is saved in `style-default_borders.stjson` and can be accessed by typing

```
. collect style use default_borders
```

as one step in building your own style.

## default\_cidelimiter

This style defines the delimiters for confidence intervals and credible intervals. It is part of the definition for the `default`, `coef-table`, and `etable` styles and is defined using the following commands:

```
. collect style clear
. collect style cell, basestyle cidelimiter(" ") cridelimiter(" ")
```

This style is saved in `style-default_cidelimiter.stjson` and can be accessed by typing

```
. collect style use default_cidelimiter
```

as one step in building your own style.

## default\_halign

This style defines horizontal alignment properties. It is part of the definition for the `default`, `table`, and `tabulate` styles and is defined using the following commands:

```
. collect style clear
. collect style cell cell_type[corner], halign(center)
. collect style cell cell_type[column-header], halign(right)
. collect style cell cell_type[column-header]#result[_r_ci _r_cri], halign(center)
. collect style cell cell_type[item], halign(right)
. collect style cell cell_type[row-header], halign(left)
```

This style is saved in `style-default_halign.stjson` and can be accessed by typing

```
. collect style use default_halign
```

as one step in building your own style.

## default\_headers

This style defines header properties. It is part of the definition for the `default` style and is defined using the following commands:

```
. collect style clear
. collect style header, basestyle title(hide) level(label)
. collect style header stars, title(hide) level(hide)
. collect style row stack
```

This style is saved in `style-default_headers.stjson` and can be accessed by typing

```
. collect style use default_headers
```

as one step in building your own style.

## default\_margins

This style defines cell margin properties. It is part of the definition for the `default`, `dtable`, `etable`, `lcmstats`, `table`, `tabulate`, and `coef-table` styles and is defined using the following commands:

```
. collect style clear
. collect style cell cell_type[column-header item], margin(left, width(10pt))
```

This style is saved in `style-default_margins.stjson` and can be accessed by typing

```
. collect style use default_margins
```

as one step in building your own style.

## default\_nformats

This style defines numeric format properties. It is part of the definition for the default and coef-table styles and is defined using the following commands:

```
. collect style clear
. collect style cell, basestyle nformat(%9.0g)
. collect style cell result[_r_b _r_se _r_lb _r_ub _r_ci _r_crlb _r_crub _r_cri] ///
    , nformat(%9.0g)
. collect style cell result[_r_z _r_z_abs], nformat(%8.2f)
. collect style cell result[_r_p], nformat(%5.3f)
```

This style is saved in `style-default_nformats.stjson` and can be accessed by typing

```
. collect style use default_nformats
```

as one step in building your own style.

## default\_smcl

This style defines SMCL properties targeted to look like most tables in Stata output. It is part of the definition for the default, dtable, etable, lcstats, table, tabulate, and coef-table styles and is defined using the following commands:

```
. collect style clear
. collect style cell cell_type[item], smcl(result)
```

This style is saved in `style-default_smcl.stjson` and can be accessed by typing

```
. collect style use default_smcl
```

as one step in building your own style.

## default\_tex

This style defines L<sup>A</sup>T<sub>E</sub>X properties. It is part of the definition for the default, dtable, etable, lcstats, table, tabulate, and coef-table styles and is defined using the following commands:

```
. collect style clear
. collect style tex, begintable centering
```

This style is saved in `style-default_tex.stjson` and can be accessed by typing

```
. collect style use default_tex
```

as one step in building your own style.

## **dtable\_borders**

This style defines cell border properties. It is part of the definition for the `dtable` style and is defined using the following commands:

```
. collect style clear
. collect style cell border_block[corner], border(top)
. collect style cell border_block[column-header], border(top)
. collect style cell border_block[row-header], border(top bottom)
. collect style cell border_block[item], border(top bottom)
```

This style is saved in `style-dtable_borders.stjson` and can be accessed by typing

```
. collect style use dtable_borders
```

as one step in building your own style.

## **dtable\_composites**

This style defines some convenient composite results for use with the `dtable` command. It is not part of the definition for the `dtable` style but is described in *Composite results of [R] dtable*.

This style is defined using the following commands:

```
. collect style clear
. collect composite define idi = p10 p90, trim
. collect composite define iqi = q1 q3, trim
. collect composite define rangei = min max, trim
. collect style cell result[idi iqi rangei], sformat("%s")
```

This style is saved in `style-dtable_composites.stjson` and can be accessed by typing

```
. collect style use dtable_composites
```

as one step in building your own style.

## **dtable\_font**

This style is a placeholder intended to define cell font properties. It is part of the definition for the `dtable` style, but its definition is currently empty.

## **dtable\_halign**

This style defines horizontal alignment properties. It is part of the definition for the `dtable` style and is defined using the following commands:

```
. collect style clear
. collect style cell cell_type[column-header], halign(center)
. collect style cell cell_type[row-header], halign(left)
. collect style cell stars[label], halign(left)
```

This style is saved in `style-dtable_halign.stjson` and can be accessed by typing

```
. collect style use dtable_halign
```

as one step in building your own style.

## **dtable\_headers**

This style defines header properties. It is part of the definition for the `dtable` style and is defined using the following commands:

```
. collect style clear
. collect style header, basestyle title(hide) level(label)
. collect style row stack, nobinder
. collect style column, dups(center)
```

This style is saved in `style-dtable_headers.stjson` and can be accessed by typing

```
. collect style use dtable_headers
```

as one step in building your own style.

## **dtable\_nformats**

This style defines numeric format properties. It is part of the definition for the `dtable` style and is defined using the following commands:

```
. collect style clear
. collect style cell, basestyle nformat(%21.3fc)
. collect style cell result[N count sumw frequency fvfrequency fvrawfrequency] ///
  , nformat(%21.0fc)
. collect style cell result[percent rawpercent fvpercent fvrawpercent] ///
  , nformat(%9.1fc) sformat("(%s%")
. collect style cell result[sd], sformat("(%s)")
. collect style cell result[ regress poisson lnnormal
  kwallis kwallis_ties ///
  pearson fisher lrchi2 ///
  svylr svywald svyllwald ///
  ], nformat(%6.3f) minimum(.001)
```

This style is saved in `style-dtable_nformats.stjson` and can be accessed by typing

```
. collect style use dtable_nformats
```

as one step in building your own style.

## **etable\_borders**

This style defines cell border properties. It is part of the definition for the `etable` style and is defined using the following commands:

```
. collect style clear
. collect style cell border_block[corner], border(top)
. collect style cell border_block[column-header], border(top)
. collect style cell border_block[row-header], border(top bottom)
. collect style cell border_block[item], border(top bottom)
```

This style is saved in `style-etable_borders.stjson` and can be accessed by typing

```
. collect style use etable_borders
```

as one step in building your own style.

## etable\_etable

This style defines properties unique to the `etable` command; see [R] [etable](#). These properties control the default behavior for the `column()`, `cstat()`, `eqrcode()`, `equations()`, `fvlabel`, `keep()`, `mstat()`, `showeq`, `showstars`, `showstarsnote`, and `varlabel` options. It is part of the definition for the `etable` style.

This style is saved in `style-etable_etable.stjson` and can be accessed by typing

```
. collect style use etable_etable
```

as one step in building your own style.

## etable\_font

This style is a placeholder intended to define cell font properties. It is part of the definition for the `etable` style, but its definition is currently empty.

## etable\_halign

This style defines horizontal alignment properties. It is part of the definition for the `etable` style and is defined using the following commands:

```
. collect style clear
. collect style cell cell_type[column-header], halign(center)
. collect style cell cell_type[row-header], halign(left)
. collect style cell stars[label], halign(left)
```

This style is saved in `style-etable_halign.stjson` and can be accessed by typing

```
. collect style use etable_halign
```

as one step in building your own style.

## etable\_headers

This style defines header properties. It is part of the definition for the `etable` style and is defined using the following commands:

```
. collect style clear
. collect style header, basestyle title(hide) level(value)
. collect style header cmdset, level(label)
. collect style header stars, level(hide)
. collect style header result, level(label)
. collect style header result[ _r_b _r_se ///
                             _r_z _r_z_abs ///
                             _r_p ///
                             _r_lb _r_ub _r_ci ///
                             _r_crlb _r_crub _r_cri ///
                           ] , level(hide)
. collect style row stack, nobinder
. collect style column, dups(center)
```

This style is saved in `style-etable_headers.stjson` and can be accessed by typing

```
. collect style use etable_headers
```

as one step in building your own style.

## etable\_nformats

This style defines numeric format properties. It is part of the definition for the `etable` style and is defined using the following commands:

```
. collect style clear
. collect style cell, basestyle nformat(%9.2f)
. collect style cell result[N df_m df_r], nformat(%21.0f)
. collect style cell result[ _r_b _r_se ///
                        _r_lb _r_ub _r_ci ///
                        _r_crlb _r_crub _r_cri ///
                    ] , nformat(%9.3f)
. collect style cell result[_r_p] , nformat(%9.2f)
. collect style cell result[_r_se]#stars[value], sformat("(%s)")
. collect style cell result[_r_ci _r_cri]#stars[value], sformat("[%s]")
```

This style is saved in `style-etable_nformats.stjson` and can be accessed by typing

```
. collect style use etable_nformats
```

as one step in building your own style.

## etable\_showitem

This style defines `showbase`, `showomit`, and `showempty` properties. It is part of the definition for the `etable` style and is defined using the following commands:

```
. collect style clear
. collect style showbase off
. collect style showempty off
. collect style showomit off
```

This style is saved in `style-etable_showitem.stjson` and can be accessed by typing

```
. collect style use etable_showitem
```

as one step in building your own style.

## etable\_stars

This style defines stars properties. It is part of the definition for the `etable` style and is defined using the following commands:

```
. collect style clear
. collect stars _r_p .05 "*" .01 "***", attach(_r_b) dimension
```

This style is saved in `style-etable_stars.stjson` and can be accessed by typing

```
. collect style use etable_stars
```

as one step in building your own style.

## lcstats\_font

This style is a placeholder intended to define cell font properties. It is part of the definition for the `lcstats` style, but its definition is currently empty.

## lcstats\_halign

This style defines horizontal alignment properties. It is part of the definition for the `lcstats` style and is defined using the following commands:

```
. collect style clear
. collect style cell cell_type[column-header], halign(right)
. collect style cell cell_type[row-header], halign(right)
```

This style is saved in `style-lcstats_halign.stjson` and can be accessed by typing

```
. collect style use lcstats_halign
```

as one step in building your own style.

## lcstats\_headers

This style defines header properties. It is part of the definition for the `lcstats` style and is defined using the following commands:

```
. collect style clear
. collect style header, basestyle title(hide) level(label)
. collect style row stack, nobinder
. collect style column, dups(center) extraspace(1)
```

This style is saved in `style-lcstats_headers.stjson` and can be accessed by typing

```
. collect style use lcstats_headers
```

as one step in building your own style.

## lcstats\_nformats

This style defines numeric format properties. It is part of the definition for the `lcstats` style and is defined using the following commands:

```
. collect style clear
. collect style cell, basestyle nformat(%9.0gc)
. collect style cell result[N k_classes rank df], nformat(%21.0fc)
. collect style cell result[entropy], nformat(%6.4f)
. collect style cell result[mean sd], nformat(%21.2fc)
. collect style cell result[aic aicc bic caic], nformat(%21.2fc)
. collect style cell result[l1 lmr vlmr], nformat(%21.2fc)
. collect style cell result[p_lmr p_vlmr], nformat(%6.3f) minimum(.001)
```

This style is saved in `style-lcstats_nformats.stjson` and can be accessed by typing

```
. collect style use lcstats_nformats
```

as one step in building your own style.

## lcstats\_smcl

This style defines *SMCL* properties. It is part of the definition for the *lcstats* style and is defined using the following commands:

```
. collect style clear
. collect style cell cell_type[item], smcl(result)
. collect style smcl, wrapnotes
```

This style is saved in `style-lcstats_smcl.stjson` and can be accessed by typing

```
. collect style use lcstats_smcl
```

as one step in building your own style.

## lcstats\_txt

This style defines plain text properties. It is part of the definition for the *lcstats* style and is defined using the following commands:

```
. collect style clear
. collect style txt, wrapnotes
```

This style is saved in `style-lcstats_txt.stjson` and can be accessed by typing

```
. collect style use lcstats_txt
```

as one step in building your own style.

**table\_cidelimiter**

This style defines the delimiters for confidence intervals and credible intervals. It is part of the definition for the table style and is defined using the following commands:

```
. collect style clear
. collect style cell, basestyle cidelimiter(" ") cridelimiter(" ")
```

This style is saved in `style-table_cidelimiter.stjson` and can be accessed by typing

```
. collect style use table_cidelimiter
```

as one step in building your own style.

**table\_headers**

This style defines header properties. It is part of the definition for the table style and is defined using the following commands:

```
. collect style clear
. collect style header, basestyle title(label) level(label)
. collect style column, dups(center) extraspace(1)
. collect style header across, title(hide)
. collect style header coleq, title(hide)
. collect style header colname, title(hide)
. collect style header command, title(hide)
. collect style header result, title(hide)
. collect style header roweq, title(hide)
. collect style header rowname, title(hide)
. collect style header stars, title(hide) level(hide)
. collect style header statcmd, title(hide)
. collect style header var, title(hide)
. collect style row stack
```

This style is saved in `style-table_headers.stjson` and can be accessed by typing

```
. collect style use table_headers
```

as one step in building your own style.

## table\_nformats

This style defines numeric format properties similar to `default_nformats` but adds numeric formats for targeted statistics of the `table` command. It is part of the definition for the `table` style and is defined using the following commands:

```
. collect style clear
. collect style cell, basestyle nformat(%9.0g)
. collect style cell result[_r_b _r_se _r_lb _r_ub _r_ci _r_crlb _r_crub _r_cri] ///
    , nformat(%9.0g)
. collect style cell result[_r_z _r_z_abs], nformat(%8.2f)
. collect style cell result[_r_p], nformat(%5.3f)
. collect style cell result[count frequency fvrawfrequency], nformat(%18.0fc)
. collect style cell result[fvpercent fvrawpercent percent rawpercent] ///
    , nformat(%9.2f)
. collect style cell result[fvproportion fvrawproportion proportion rawproportion] ///
    , nformat(%9.4g)
. collect style cell result[rawtotal sumw total fvfrequency], nformat(%18.6gc)
```

This style is saved in `style-table_nformats.stjson` and can be accessed by typing

```
. collect style use table_nformats
```

as one step in building your own style.

## tabulate\_borders

This style defines cell border properties. It is part of the definition for the `tabulate` style and is defined using the following commands:

```
. collect style clear
. collect style cell border_block[corner], border(top)
. collect style cell border_block[column-header], border(top)
. collect style cell border_block[row-header], border(top bottom)
. collect style cell border_block[item], border(top bottom)
. collect style cell border_block[row-header]#item_type[overall] ///
    , border(top bottom, pattern(none))
. collect style cell border_block[item]#item_type[overall] ///
    , border(top bottom, pattern(none))
```

This style is saved in `style-tabulate_borders.stjson` and can be accessed by typing

```
. collect style use tabulate_borders
```

as one step in building your own style.

## tabulate\_headers

This style defines header properties. It is part of the definition for the `tabulate` style and is defined using the following commands:

```
. collect style clear
. collect style header, basestyle title(label) level(label)
. collect style header result, title(hide)
. collect style header _overall item_type, title(hide) level(hide)
. collect style column, dups(center) extraspace(1)
. collect style row stack
```

This style is saved in `style-tabulate_headers.stjson` and can be accessed by typing

```
. collect style use tabulate_headers
```

as one step in building your own style.

## tabulate\_nformats

This style defines numeric format properties. It is part of the definition for the `tabulate` style and is defined using the following commands:

```
. collect style clear
. collect style cell, basestyle nformat(%18.0g)
. collect style cell result[frequency count], nformat(%18.0fc)
. collect style cell result[mean sd sumw expected cchi2 clrchi2], nformat(%18.1fc)
. collect style cell result[ CramersV ///
    ase_gam ase_taub ///
    chi2 chi2_lr ///
    gamma taub ///
    F_Pear F_Penl ///
    cun_Pear cun_Penl ///
    cun_LR cun_LRnl ///
    F_LR F_LRnl ///
    F_Wald F_LLW ///
    Fun_Wald Fun_LLW ///
    cun_Wald cun_LLW ///
    mgdeff cvgdeff ///
    ], nformat(%18.4fc)
. collect style cell result[ p_p_exact p1_exact p_lr ///
    p_Pear p_Penl ///
    p_LR p_LRnl ///
    p_Wald p_LLW ///
    pun_Wald pun_LLW ///
    ], nformat(%18.4fc) minimum(.0001)
. collect style cell result[proportion row_proportion column_proportion] ///
    , nformat(%18.4fc)
. collect style cell result[ row column cell ///
    percent cumulative ///
    row_percent column_percent ///
    ], nformat(%9.2f)
. collect style cell result[deff deft] , nformat(%9.3f)
. collect style cell result[ cv ///
    df1_Pear df2_Pear ///
    df1_Penl df2_Penl ///
    df1_LR df2_LR ///
    df1_LRnl df2_LRnl ///
    ], nformat(%18.2f)
. collect style header cmdset, title(hide) level(hide)
. collect style header var, title(hide) level(label)
```

This style is saved in `style-tabulate_nformats.stjson` and can be accessed by typing

```
. collect style use tabulate_nformats
```

as one step in building your own style.

## Modifying the default style

If you routinely change your style to one of the styles installed with Stata or to one you have created, you can consider changing the style used by default. For information on changing the default style used by `collect`, see [TABLES] [set collect\\_style](#). For information on changing the default style used by `etable`, see [TABLES] [set etable\\_style](#). For information on changing the default style used by `dtable`, see [TABLES] [set dtable\\_style](#). For information on changing the default style used by `table`, see [TABLES] [set table\\_style](#). For information on changing the default style used by `tabulate` and `svy: tabulate`, see [TABLES] [set tabulate\\_style](#).

## Also see

- [TABLES] **collect style use** — Use collection styles from disk
- [TABLES] **set collect\_style** — Style settings for collections
- [TABLES] **set dtable\_style** — Default style settings for dtable
- [TABLES] **set etable\_style** — Default style settings for etable
- [TABLES] **set table\_style** — Default style settings for table
- [TABLES] **set tabulate\_style** — Default style settings for tabulate

Stata, Stata Press, Mata, NetCourse, and NetCourseNow are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow is a trademark of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2025 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the FAQ on [citing Stata documentation](#).